

6 esercizi su problemi e modelli di ottimizzazione

Esercizio 1. Un pasticcere deve decidere quali torte preparare. Il listino prevede un insieme T di possibili torte diverse, ma il pasticcere può decidere di non preparare alcuni tipi. Sappiamo che una torta di tipo $t \in T$ viene venduta al prezzo di p_t euro e che non vengono mai preparate due torte dello stesso tipo. È possibile vendere una torta in confezione di lusso: in questo caso il prezzo della torta t aumenta di f_t euro. Sono disponibili complessivamente n confezioni di lusso. Per preparare ogni torta $t \in T$ sono necessari alcuni ingredienti che sono descritti dall'insieme I . La quantità di ingrediente $i \in I$ necessaria per preparare la torta t è pari a q_{it} unità. Gli ingredienti vengono acquistati in confezioni: una confezione dell'ingrediente i costa c_i e contiene d_i unità dell'ingrediente.

- a) Scrivere un modello di Programmazione Lineare Intera per aiutare il pasticcere a decidere quali torte preparare e come approvvigionarsi degli ingredienti, in modo da massimizzare il profitto netto della vendita dei dolci, sapendo che tutti i dolci prodotti vengono venduti e che il ricavo della vendita, escludendo gli aumenti per le confezioni di lusso, deve essere almeno pari a R euro.
- b) Scrivere nella sintassi di AMPL il modello di PLI formulato al punto a).

Esercizio 2. Una raffineria deve miscelare 4 tipi di petrolio grezzo per produrre 2 tipi di benzina (super95 e super100). La tabella seguente indica quanti barili di ogni tipo di petrolio grezzo sono a disposizione della raffineria ed il relativo costo.

Tipo di petrolio grezzo	Disponibilità (barili)	Costo (euro/barile)
A	5000	9
B	2400	7
C	4000	12
D	1500	6

Ogni benzina deve soddisfare delle specifiche tecniche che riguardano la quantità di ogni petrolio grezzo usata per la sua produzione. In particolare, sappiamo che al massimo il 30% della benzina super95 può essere costituito dal petrolio grezzo di tipo D e almeno il 40% della benzina super100 deve essere costituito dal petrolio grezzo di tipo C. Le benzine super95 e super100 vengono vendute rispettivamente ad un prezzo di 15 e 18 euro/barile.

Sapendo che tutta la benzina prodotta sarà venduta, la raffineria deve decidere quanti barili di benzine super95 e super100 produrre e come miscelare i diversi tipi di petrolio grezzo per produrre le benzine, in modo tale da massimizzare il profitto totale (differenza tra ricavo e spesa).

- a) Formulare un modello di Programmazione Lineare Intera per aiutare il responsabile della raffineria a risolvere il problema.
- b) Scrivere nella sintassi di AMPL il modello formulato al punto a).

Esercizio 3. Un'azienda produttrice di automobili deve pianificare lo smistamento delle vetture in un paese straniero. Le vetture raggiungono il paese via mare e le navi possono attraccare in 3 diversi porti, P_1 , P_2 e P_3 . Per ciascun porto l'azienda deve pagare una tassa per ciascuna automobile del carico: nel porto P_1 la tassa è di 250 euro a vettura, nel porto P_2 è di 150 euro a vettura, mentre nel porto P_3 è di 200 euro a vettura. Le automobili devono essere inviate dai porti ai 4 centri di smistamento presenti nel paese, S_1 , S_2 , S_3 e S_4 . Il costo (in euro) per inviare una vettura da ogni porto ad ogni centro di smistamento è indicato nella tabella seguente:

Porto	Centri di smistamento			
	S_1	S_2	S_3	S_4
P_1	15	25	10	20
P_2	40	30	30	50
P_3	25	35	20	30

La richiesta del centro di smistamento S_1 è di almeno 170 vetture, quella di S_2 è di almeno 130 vetture, quella di S_3 è di almeno 100 vetture, mentre quella di S_4 è di almeno 200 vetture. Per motivi strategici, il centro S_3 deve essere rifornito da un solo porto. Inoltre, se il porto P_2 serve il centro di smistamento S_2 , allora deve servire anche il centro S_4 . È noto che il porto P_1 può gestire al massimo 350 vetture, il porto P_2 al massimo 200 vetture, mentre il porto P_3 al massimo 300 vetture.

L'azienda deve decidere come distribuire le vetture dai porti verso i centri di smistamento, rispettando le richieste dei centri e le capacità dei porti, in modo da minimizzare il costo totale dato dai costi di trasporto più le tasse pagate nei porti.

- Formulare un modello di Programmazione Lineare Intera per aiutare l'azienda a risolvere il problema.
- Scrivere nella sintassi di AMPL il modello formulato al punto a).

Esercizio 4. Un'azienda che produce automobili ha a disposizione 4 impianti produttivi. A causa delle diverse tecnologie e del personale presenti negli impianti, il costo di produzione di un'automobile è diverso in ogni impianto ed è indicato nella tabella seguente. Nella stessa tabella sono indicati anche il numero di ore di lavoro ed il numero di tonnellate di acciaio necessarie per produrre un'automobile in ogni impianto.

Impianti	Costo di produzione (€)	Lavoro (ore)	Acciaio (ton)
1	15000	8	3
2	10000	9	4
3	9000	7	5
4	7000	5	6

Ogni mese l'azienda ha a disposizione complessivamente 13000 ore di lavoro e 4000 tonnellate di acciaio per la produzione nei 4 impianti. L'azienda deve decidere il piano di produzione del prossimo mese con l'obiettivo di minimizzare i costi di produzione, sapendo che dovrà produrre complessivamente almeno 1000 automobili e almeno 400 automobili dovranno essere prodotte nell'impianto 3.

- Formulare un modello di Programmazione Lineare Intera per aiutare l'azienda a risolvere il problema.
- Scrivere nella sintassi di AMPL il modello formulato al punto a).

Esercizio 5. Una compagnia di assicurazioni ha deciso di aprire p filiali per servire n clienti. Il volume di lavoro portato da ogni cliente i richiede d_i ore di lavoro ogni mese. Ogni dipendente della compagnia lavora L ore al mese e ogni filiale j deve avere un numero di dipendenti compreso tra m_j e M_j . La compagnia deve decidere quanti dipendenti assumere in ogni filiale e come assegnare i clienti alle filiali (ogni cliente deve essere assegnato ad una sola filiale), in modo da soddisfare le richieste dei clienti e la numerosità dei dipendenti nelle filiali, con l'obiettivo di assumere il numero minimo possibile di dipendenti.

- a) Formulare un modello di Programmazione Lineare Intera per aiutare il responsabile della compagnia a risolvere il problema.
- b) Scrivere nella sintassi di AMPL il modello formulato al punto a).

Esercizio 6. L'amministrazione di una regione deve pianificare l'apertura di asili nido in un'area montana. Nell'area sono presenti alcuni paesi, descritti dall'insieme I , dove abitano le famiglie dei bambini a cui deve essere garantito il servizio di asilo nido. Gli asili nido possono essere costruiti in qualsiasi paese $i \in I$, ma il budget che lo Stato mette a disposizione permette di costruire al massimo p asili. In ogni paese $i \in I$ ci sono b_i bambini a cui deve essere garantito il servizio. Tutti i bambini di un paese devono essere assegnati allo stesso asilo nido e, se l'asilo nido di riferimento non si trova nello stesso paese, viene attivato un servizio di scuolabus tra il paese dove vivono i bambini e l'asilo nido. Il servizio di scuolabus per portare tutti i bambini del paese i al nido aperto nel paese j costa c_{ij} euro all'anno. Oltre ai costi del servizio di scuolabus, l'amministrazione deve considerare anche i costi del personale: assumere un educatore costa g euro all'anno. Inoltre, la normativa vigente richiede che ci sia un educatore ogni q bambini.

L'amministrazione deve decidere in quali paesi aprire gli asili nido, quanti educatori assumere in ogni nuovo asilo nido e come assegnare i bambini di ogni paese ad un nuovo asilo nido, con l'obiettivo di minimizzare i costi annuali di gestione (costi di trasporto più costi del personale).

- a) Formulare un modello di Programmazione Lineare Intera per aiutare il responsabile dell'amministrazione a risolvere il problema.
- b) Scrivere nella sintassi di AMPL il modello formulato al punto a).

Soluzioni

Esercizio 1.

a) Definiamo le seguenti variabili:

$$x_t = \begin{cases} 1 & \text{se il pasticcere decide di preparare la torta } t \\ 0 & \text{altrimenti} \end{cases} \quad \text{per ogni } t \in T;$$

$$y_t = \begin{cases} 1 & \text{se la torta } t \text{ viene venduta in una confezione di lusso} \\ 0 & \text{altrimenti} \end{cases} \quad \text{per ogni } t \in T;$$

$z_i \in \mathbb{N}$ indica il numero di confezioni acquistate dell'ingrediente i , per ogni $i \in I$.

Il problema si può formulare mediante il seguente modello di PLI:

$$\max \sum_{t \in T} p_t x_t + \sum_{t \in T} f_t y_t - \sum_{i \in I} c_i z_i \quad (1)$$

$$y_t \leq x_t \quad \forall t \in T \quad (2)$$

$$\sum_{t \in T} y_t \leq n \quad (3)$$

$$\sum_{t \in T} q_{it} x_t \leq d_i z_i \quad \forall i \in I \quad (4)$$

$$\sum_{t \in T} p_t x_t \geq R \quad (5)$$

$$x_t \in \{0, 1\} \quad \forall t \in T$$

$$y_t \in \{0, 1\} \quad \forall t \in T$$

$$z_i \in \mathbb{Z}, \quad z_i \geq 0 \quad \forall i \in I$$

Nella funzione obiettivo il primo termine rappresenta il ricavo dovuto alla vendita delle torte, il secondo il ricavo dovuto alle eventuali confezioni di lusso, il terzo i costi di acquisto degli ingredienti.

I vincoli (2) impongono che una torta può essere venduta in una confezione di lusso solo se viene effettivamente preparata. I vincoli (3) impongono che si possono preparare al massimo n confezioni di lusso. I vincoli (4) impongono che, ogni ingrediente i , la quantità di ingrediente i necessaria per preparare tutte le torte non può superare la quantità di ingrediente i che è stata acquistata. Il vincolo (5) impone che il ricavo dovuto alla vendita delle torte sia almeno pari a R euro.

b) Il modello di PLI formulato al punto a) può essere scritto nella sintassi di AMPL nel modo seguente:

Insiemi e parametri

```
set T;
set I;
param p{t in T};
param f{t in T};
param c{i in I};
param d{i in I};
param q{i in I, t in T};
param R;
```

variabili

```
var x{t in T} binary;
```

```
var y{t in T} binary;
var z{i in I} integer, >=0;

# funzione obiettivo

maximize profitto: sum{t in T} p[t]*x[t] + sum{t in T} f[t]*y[t] - sum{i in I} c[i]*z[i];

# vincoli

s.t. conf_lusso1{t in T}: y[t] <= x[t];

s.t. conf_lusso2: sum{t in T} y[t] <= n;

s.t. ingredienti{i in I}: sum{t in T} q[i,t]*x[t] <= d[i]*z[i];

s.t. ricavo: sum{t in T} p[t]*x[t] >= R;
```

Esercizio 2.

a) Definiamo le seguenti variabili:

x_1 = numero di barili di benzina super95 prodotti;

x_2 = numero di barili di benzina super100 prodotti;

y_{A1} = numero di barili di petrolio grezzo di tipo A usati per produrre la benzina super95;

y_{B1} = numero di barili di petrolio grezzo di tipo B usati per produrre la benzina super95;

y_{C1} = numero di barili di petrolio grezzo di tipo C usati per produrre la benzina super95;

y_{D1} = numero di barili di petrolio grezzo di tipo D usati per produrre la benzina super95;

y_{A2} = numero di barili di petrolio grezzo di tipo A usati per produrre la benzina super100;

y_{B2} = numero di barili di petrolio grezzo di tipo B usati per produrre la benzina super100;

y_{C2} = numero di barili di petrolio grezzo di tipo C usati per produrre la benzina super100;

y_{D2} = numero di barili di petrolio grezzo di tipo D usati per produrre la benzina super100;

Il problema si può formulare mediante il seguente modello di PLI:

$$\begin{aligned} \max \quad & 15x_1 + 18x_2 - 9(y_{A1} + y_{A2}) - 7(y_{B1} + y_{B2}) \\ & - 12(y_{C1} + y_{C2}) - 6(y_{D1} + y_{D2}) \end{aligned}$$

$$x_1 = y_{A1} + y_{B1} + y_{C1} + y_{D1} \quad (6)$$

$$x_2 = y_{A2} + y_{B2} + y_{C2} + y_{D2} \quad (7)$$

$$y_{A1} + y_{A2} \leq 5000 \quad (8)$$

$$y_{B1} + y_{B2} \leq 2400 \quad (9)$$

$$y_{C1} + y_{C2} \leq 4000 \quad (10)$$

$$y_{D1} + y_{D2} \leq 1500 \quad (11)$$

$$y_{D1} \leq 0.3x_1 \quad (12)$$

$$y_{C2} \geq 0.4x_2 \quad (13)$$

$$x_1, x_2 \in \mathbb{Z}, \quad x_1, x_2 \geq 0$$

$$y_{A1}, y_{B1}, y_{C1}, y_{D1}, y_{A2}, y_{B2}, y_{C2}, y_{D2} \geq 0$$

La funzione obiettivo è la differenza tra il ricavo dovuto alla vendita dei due tipi di benzina ed il costo dei quattro tipi di petrolio grezzo utilizzati per la produzione delle benzine. I vincoli (6)–(7) impongono che le benzine siano prodotte utilizzando i quattro tipi di petrolio grezzo. I vincoli (8)–(11) riguardano la disponibilità dei quattro tipi di petrolio grezzo. I vincoli (12)–(13) riguardano le specifiche tecniche che devono soddisfare i due tipi di benzina.

b) Il modello di PLI formulato al punto a) può essere scritto nella sintassi di AMPL nel modo seguente:

```
var x1 integer >=0;
```

```
var x2 integer >=0;
```

```
var yA1 >=0 ;
```

```
var yB1 >=0 ;
```

```
var yC1 >=0 ;
```

```
var yD1 >=0 ;
```

```
var yA2 >=0 ;
```

```
var yB2 >=0 ;
```

```
var yC2 >=0 ;
```

```
var yD2 >=0 ;
```

```
maximize profit: 15*x1 + 18*x2 - 9*(yA1+yA2) - 7*(yB1+yB2) - 12*(yC1+yC2) - 6*(yD1+yD2);
```

```
s.t. v1: x1 = yA1 + yB1 + yC1 + yD1;
```

```
s.t. v2: x2 = yA2 + yB2 + yC2 + yD2;
```

s.t. d1: $yA1 + yA2 \leq 5000$;

s.t. d2: $yB1 + yB2 \leq 2400$;

s.t. d3: $yC1 + yC2 \leq 4000$;

s.t. d4: $yD1 + yD2 \leq 1500$;

s.t. m1: $yD1 \leq 0.3 \cdot x1$;

s.t. m2: $yC2 \geq 0.4 \cdot x2$;

Esercizio 3.

a) Parametri:

c_{ij} = costo per inviare una vettura dal porto P_i al centro di smistamento S_j (per ogni $i = 1, 2, 3$, $j = 1, \dots, 4$),

t_i = tassa da pagare per ogni vettura che transita nel porto P_i (per ogni $i = 1, 2, 3$),

r_j = numero di vetture richieste dal centro di smistamento S_j (per ogni $j = 1, \dots, 4$),

M_i = massimo numero di vetture che può gestire il porto P_i (per ogni $i = 1, 2, 3$).

Variabili:

$x_{ij} = \begin{cases} 1 & \text{se il porto } P_i \text{ serve il centro di smistamento } S_j \\ 0 & \text{altrimenti} \end{cases}$ per ogni $i = 1, 2, 3$, $j = 1, \dots, 4$,

y_{ij} = numero di vetture spedite dal porto P_i al centro di smistamento S_j (per ogni $i = 1, 2, 3$, $j = 1, \dots, 4$).

Il problema si può formulare mediante il seguente modello di PLI:

$$\begin{aligned} \min \quad & \sum_{i=1}^3 \sum_{j=1}^4 c_{ij} y_{ij} + \sum_{i=1}^3 t_i \sum_{j=1}^4 y_{ij} \\ & \sum_{i=1}^3 y_{ij} \geq r_j \quad \forall j = 1, \dots, 4 \end{aligned} \quad (14)$$

$$\sum_{j=1}^4 y_{ij} \leq M_i \quad \forall i = 1, \dots, 3 \quad (15)$$

$$y_{ij} \leq M_i x_{ij} \quad \forall i = 1, \dots, 3, j = 1, \dots, 4 \quad (16)$$

$$\sum_{i=1}^3 x_{i3} = 1 \quad (17)$$

$$x_{22} \leq x_{24} \quad (18)$$

$$x_{ij} \in \{0, 1\}$$

$$y_{ij} \in \mathbb{Z}, y_{ij} \geq 0$$

La funzione obiettivo è la somma dei costi di trasporto dai porti ai centri di smistamento e delle tasse pagate nei porti.

I vincoli (14) impongono che siano soddisfatte le richieste dei centri di smistamento.

I vincoli (15) impongono che siano rispettate le capacità dei porti

I vincoli (16) legano le variabili x e y : se il porto P_i non serve il centro di smistamento S_j (cioè $x_{ij} = 0$), allora nessuna vettura può essere spedita dal porto P_i al centro S_j . Altrimenti, se il porto P_i serve il centro di smistamento S_j (cioè $x_{ij} = 1$), il numero massimo di vetture spedite è minore o uguale alla capacità M_i del porto P_i .

Il vincolo (17) impone che il centro S_3 sia servito da un solo porto.

Il vincolo (18) impone che se il porto P_2 serve il centro S_2 , allora deve servire anche il centro S_4 .

b) Il modello formulato al punto a) può essere scritto nella sintassi di AMPL utilizzando i seguenti file:

File del modello (es1.mod):

```
#----- parametri -----  
  
param c{i in 1..3, j in 1..4};  
param t{i in 1..3};  
param r{j in 1..4};
```



```

param M{i in 1..3};

#----- variabili -----

var x{i in 1..3, j in 1..4} binary ;
var y{i in 1..3, j in 1..4} integer >= 0 ;

#----- funzione obiettivo -----

minimize costi: sum{i in 1..3, j in 1..4} c[i,j]*y[i,j]
+ sum{i in 1..3, j in 1..4} t[i]*y[i,j] ;

#----- vincoli -----

s.t. v1{j in 1..4}: sum{i in 1..3} y[i,j] >= r[j];
s.t. v2{i in 1..3}: sum{j in 1..4} y[i,j] <= M[i];
s.t. v3{i in 1..3, j in 1..4}: y[i,j] <= M[i]*x[i,j];
s.t. v4: sum{i in 1..3} x[i,3] = 1;
s.t. v5: x[2,2] <= x[2,4];

```

File dei dati (es1.dat):

```

param c: 1 2 3 4 :=
1 15 25 10 20
2 40 30 30 50
3 25 35 20 30;

```

```

param t :=
1 250
2 150
3 200;

```

```

param r :=
1 170
2 130
3 100
4 200;

```

```

param M :=
1 350
2 200
3 300;

```

Esercizio 4.

a) Introduciamo le seguenti variabili:

x_i = numero di automobili prodotte nell'impianto i (per ogni $i = 1, \dots, 4$).

Il problema si può formulare mediante il seguente modello di PLI:

$$\begin{aligned} \min \quad & 15000x_1 + 10000x_2 + 9000x_3 + 7000x_4 \\ & 8x_1 + 9x_2 + 7x_3 + 5x_4 \leq 13000 \end{aligned} \tag{19}$$

$$3x_1 + 4x_2 + 5x_3 + 6x_4 \leq 4000 \tag{20}$$

$$x_1 + x_2 + x_3 + x_4 \geq 1000 \tag{21}$$

$$x_3 \geq 400 \tag{22}$$

$$x_1, \dots, x_4 \in \mathbb{Z}, \quad x_1, \dots, x_4 \geq 0$$

La funzione obiettivo rappresenta il costo totale di produzione.

Il vincolo (19) impone che il numero totale di ore di lavoro necessarie per produrre le automobili non sia superiore al numero di ore di lavoro disponibili.

Il vincolo (20) impone che il numero totale di tonnellate di acciaio necessarie per produrre le automobili non sia superiore al numero di tonnellate di acciaio disponibili.

Il vincolo (21) impone che vengano prodotte almeno 1000 automobili.

Il vincolo (22) impone che l'impianto 3 produca almeno 400 automobili.

b) Il modello formulato al punto a) può essere scritto nella sintassi di AMPL nel modo seguente:

```
var x{i in 1..4} integer >= 0;
```

```
minimize costi: 15000*x[1] + 10000*x[2] + 9000*x[3] + 7000*x[4];
```

```
s.t. v1: 8*x[1] + 9*x[2] + 7*x[3] + 5*x[4] <= 13000;
```

```
s.t. v2: 3*x[1] + 4*x[2] + 5*x[3] + 6*x[4] <= 4000;
```

```
s.t. v3: x[1] + x[2] + x[3] + x[4] >= 1000;
```

```
s.t. v4: x[3] >= 400;
```

Esercizio 5.

a) Definiamo le seguenti variabili:

x_j = numero di dipendenti da assumere nella filiale j , per ogni $j = 1, \dots, p$;

$$y_{ij} = \begin{cases} 1 & \text{se il cliente } i \text{ è assegnato alla filiale } j, \\ 0 & \text{altrimenti,} \end{cases}$$

per ogni $i = 1, \dots, n$, $j = 1, \dots, p$.

Il problema si può formulare mediante il seguente modello di PLI:

$$\begin{aligned} \min \quad & \sum_{j=1}^p x_j \\ \sum_{j=1}^p y_{ij} &= 1 & \forall i = 1, \dots, n \end{aligned} \quad (23)$$

$$m_j \leq x_j \leq M_j \quad \forall j = 1, \dots, p \quad (24)$$

$$\sum_{i=1}^n d_i y_{ij} \leq L x_j \quad \forall j = 1, \dots, p \quad (25)$$

$$\begin{aligned} x_j &\in \mathbb{Z}, \quad x_j \geq 0 & \forall j = 1, \dots, p \\ y_{ij} &\in \{0, 1\} & \forall i = 1, \dots, n, \quad j = 1, \dots, p \end{aligned}$$

La funzione obiettivo è il numero totale di dipendenti da assumere. I vincoli (23) impongono che ogni cliente sia assegnato ad una sola filiale; (24) impongono che il numero di dipendenti da assumere in ogni filiale sia compreso nell'intervallo desiderato; (25) garantiscono che, in ogni filiale, il numero di ore di lavoro necessarie per servire i clienti assegnati a quella filiale non sia maggiore del numero totale di ore di lavoro dei dipendenti assunti nella filiale stessa.

b) Il modello di PLI formulato al punto a) può essere scritto nella sintassi di AMPL nel modo seguente:

```
param p>0;
param n>0;
param d{i in 1..n};
param L>0;
param m{j in 1..p};
param M{j in 1..p};

var x{j in 1..p} integer >=0;
var y{i in 1..n, j in 1..p} binary;

minimize NumeroDipendenti: sum{j in 1..p} x[j];

s.t. v1{i in 1..n}: sum{j in 1..p} y[i,j] = 1;

s.t. v2a{j in 1..p}: x[j] >= m[j];

s.t. v2b{j in 1..p}: x[j] <= M[j];

s.t. v3{j in 1..p}: sum{i in 1..n} d[i]*y[i,j] <= L*x[j];
```

Esercizio 6.

a) Definiamo le seguenti variabili per ogni $i, j \in I$:

$$x_i = \begin{cases} 1 & \text{se un nuovo asilo nido viene aperto nel paese } i, \\ 0 & \text{altrimenti,} \end{cases}$$

$$y_{ij} = \begin{cases} 1 & \text{se i bambini del paese } i \text{ frequenteranno l'asilo nel paese } j, \\ 0 & \text{altrimenti,} \end{cases}$$

z_i = numero di educatori che saranno assunti nel paese i .

Il problema si può formulare mediante il seguente modello di PLI:

$$\min \sum_{i \in I} \sum_{j \in I} c_{ij} y_{ij} + \sum_{i \in I} g z_i$$

$$\sum_{i \in I} x_i \leq p \quad (26)$$

$$\sum_{j \in I} y_{ij} = 1 \quad \forall i \in I \quad (27)$$

$$y_{ij} \leq x_j \quad \forall j \in I \quad (28)$$

$$\sum_{i \in I} b_i y_{ij} \leq q z_j \quad \forall j \in I \quad (29)$$

$$x_i \in \{0, 1\} \quad \forall i \in I$$

$$y_{ij} \in \{0, 1\} \quad \forall i, j \in I$$

$$z_i \in \mathbb{Z}, z_i \geq 0 \quad \forall i \in I$$

La funzione obiettivo è la somma dei costi di trasporto dello scuolabus e del personale. I vincoli (26) impongono che possono essere aperti al massimo p nuovi asili nido; (27) impongono che i bambini di ogni paese devono essere assegnati ad un asilo nido; (28) garantiscono che se i bambini del paese i vengono assegnati all'asilo nido del paese j , allora deve essere aperto l'asilo nel paese j ; (29) garantiscono che gli educatori assunti nel paese j sono sufficienti per tutti i bambini che frequenteranno l'asilo nel paese j .

b) Il modello di PLI formulato al punto a) può essere scritto nella sintassi di AMPL nel modo seguente:

```
set I;
param p>0;
param b{i in I};
param c{i in I, j in I};
param g>0;
param q>0;

var x{i in I} binary;
var y{i in I, j in I} binary;
var z{i in I} integer >=0;

minimize Costo: sum{i in I, j in I} c[i,j]*y[i,j] + sum{i in I} g*z[i];

s.t. v1: sum{i in I} x[i] <= p;
s.t. v2{i in I}: sum{j in I} y[i,j] = 1;
s.t. v3{j in I}: y[i,j] <= x[j];
s.t. v4{j in I}: sum{i in I} b[i]*y[i,j] <= q*z[j];
```